
11 olimpiada

Jaunesniųjų grupė


173. Netikra moneta. Monetas sunumeruojame nuo 1 iki 10. Netikros monetos ieškojimas pavaizduotas paveikslėlyje. Šis sprendimas nėra vienintelis. Iš jo matome, kad turint 10 monetų užteks trijų ėjimų. 

174. Video įrašai. Kadangi tą pačią laidą galime skaidyti ir įrašyti į kelias vaizdajuostes, laidų skaičius bei konkrečių laidų trukmės nėra svarbios. Reikalinga tik suminė visų laidų trukmė. Tą trukmę patogiausia skaičiuoti minutėmis. Savaitėje yra 7 ( 24 ( 60 ( 10080 minučių, tad laidų trukmei minutėmis vaizduoti tikrai užteks integer tipo.


Tarkime, visos laidos trunka val valandų ir min minučių. Tuomet jų trukmė minutėmis lygi:


trukmė ( val ( 60 + min 


Laidoms įrašyti reikalingų vaizdajuosčių skaičius lygus:


skaičius ( trukmė div 240,        jei trukmė mod 240 ( 0


                 trukmė div 240 + 1,  jei trukmė mod 240 ( 0


Jei perkame 240 minučių trukmės vaizdajuostę, vienos minutės įrašo kaina lygi 1090 cnt/240 min ( 4,54 cnt/min. Jei perkame 180 minučių vaizdajuostę – 915 cnt/180 min ( 5,08 cnt/min. 


Akivaizdu, kad būtina pirkti trukmė div 240 vaizdajuosčių, kurių trukmė 240 minučių. Jei likutis (trukmė mod 240) netelpa į vieną 180 minučių vaizdajuostę, tuomet perkama dar viena 240 minučių trukmės vaizdajuostė, jei telpa – užteks ir trumpesnės vaizdajuostės. 


program vaizdajuostės;


  const TR_ILG = 240;  { ilgesnių vaizdajuosčių trukmė }


       TR_TRU = 180;  { trumpesnių vaizdajuosčių trukmė }


       KAINA_ILG_LT = 10; { ilgesnių vaizdajuosčių kainos dalis litais }


       KAINA_ILG_CT = 90; { ilgesnių vaizdajuosčių kainos dalis centais }


       KAINA_TRU_LT = 9;  { trumpesnių vaizdajuosčių kainos dalis litais }


       KAINA_TRU_CT = 15; { trumpesnių vaizdajuosčių kainos dalis centais }





  var n,              { laidų skaičius }


      trukmė,         { suminė laidų trukmė }


      ilgos, trumpos, { reikalingų ilgų ir trumpų vaizdajuosčių skaičiai }


      litai, centai,  { kaina litais bei centais }


      likutis, i, val, min: integer;


begin


  readln (n);


  trukmė := 0;


  { skaitome duomenis ir iš karto sumuojame laidų trukmes }


  for i := 1 to n do


    begin


      readln (val, min);


      trukmė := trukmė + (val * 60 + min)


    end;


  { randame reikalingą vaizdajuosčių kiekį }


  ilgos := trukmė div TR_ILG;


  trumpos := 0;


  likutis :( trukmė mod TR_ILG;


  if likutis > 0


     then if likutis <= TR_TRU


             then trumpos := trumpos + 1


             else ilgos := ilgos + 1;


  { apskaičiuojame reikalingą pinigų sumą }


  litai :=  KAINA_ILG_LT * ilgos + KAINA_TRU_LT * trumpos;


  centai := KAINA_ILG_CT * ilgos + KAINA_TRU_CT * trumpos;


  litai := litai + centai div 100;


  centai := centai mod 100;


  { spausdiname rezultatą }


  writeln (litai, 'lt. ', centai, 'cnt.');


end.








175. Plytelių dėliojimas. Pirmiausia reikia rasti didžiausią teigiamą skaičių sk, kurio kvadratas neviršytų n, t. y. galiotų nelygybė sk2 ( n. To ieško funkcija neviršija:





function neviršija (n: integer): integer;


  { randa didžiausią skaičių, kurio kvadratas neviršija n }


  var sk: integer;


begin


  sk := n div 2;


  while sk > n div sk do


    sk := sk - 1;


  neviršija := sk;


end; { neviršija }





O tada mažindami sk iš n atiminėjame skaičių kvadratus kol n tampa lygus nuliui. Pateikiame visą programos tekstą:


program kaladėlių_dėliojimas;


  var n, sk, kvadratas: integer;


begin


  readln (n);


  { rasime didžiausią skaičių, kurio kvadratas neviršija n }


  sk := neviršija (n);


  while n > 0 do


    begin


      kvadratas := sk * sk;


      if kvadratas <= n


       then begin


              writeln (kvadratas);


              n := n - kvadratas;


            end


       else sk := sk - 1;


    end;


end.





Vyresniųjų grupė


176. Netikra moneta. a) Monetas sunumeruojame nuo 1 iki 10. Netikros monetos ieškojimas pavaizduotas kitame lape esančiame paveikslėlyje. Šis sprendimas nėra vienintelis. Iš jo matome, kad turint 10 monetų užteks trijų ėjimų. 

b) Jei turime n monetų kiekvieną kartą monetas dalijame į tris krūveles. Jei monetų skaičius nedalus iš trijų, tuomet likusias monetas (ar monetą) padedame į skirtingas krūveles. Po tokio padalijimo bent dviejose krūvelėse monetų skaičius būtinai yra vienodas. 

Tuomet pasirenkame dvi lygias krūveles ir jas sveriame. Jei viena krūvelė sunkesnė už kitas – netikra moneta yra sunkesnėje krūvelėje, jei abi krūvelės vienodo svorio, netikra moneta yra atidėtoje krūvelėje. Taigi, vienu svėrimu dviem trečdaliais sumažiname nepatikimų monetų skaičių. 


Veiksmus kartojame kol nepatikimų monetų krūvelė sumažėja iki vienos monetos. Ji ir yra ieškomoji netikra moneta.


program netikra_moneta;


  var n, svėrimai: integer;


begin


  readln (n);


  svėrimai := 0;


  while n > 3 do


    begin


      svėrimai := svėrimai + 1;


      if n mod 3 = 0


         then n := n div 3


         else n := n div 3 + 1;


    end;


  { mažiau nei keturioms monetoms užtenka vieno svėrimo }


  svėrimai := svėrimai + 1;


  writeln (svėrimai);


end.








177. Gaukite skaičių. Pateikiame vieną iš kelių galimų uždavinio sprendimo būdų. Formuojame visus galimus reiškinius, kuriuos galime gauti iš devynių skaitmenų. Tą darome rekursiškai po kiekvieno skaitmens paeiliui įrašydami pliusą (+), minusą (–) ir neįterpdami jokio ženklo. 


Turėdami reiškinį apskaičiuojame jo reikšmę ir patikriname ar ji sutampa su duotuoju skaičiumi. Kadangi reiškinyje yra tik sudėties ir atimties operacijos, jo reikšmę galime skaičiuoti iš kairės į dešinę.


program gaukite_skaičių;





  const MAX = 9; { duotų skaitmenų skaičius }


  type operacijos = (plius, minus, nieko);


       lentelė = array [1..MAX] of operacijos;





  var n: longint;     { duotasis skaičius }


      oper: lentelė;  { operacijų lentelė }


      rasta: boolean; { ar rastas nors vienas sprendinys }





  function skaičiuoti (oper: lentelė): longint;


    { apskaičiuoja reiškinio reikšmę }


    var i, reikšmė, dėmuo, ženklas: longint;


  begin


    reikšmė := 0;


    dėmuo := 1; ženklas := 1;


    for i := 2 to MAX do


      if oper[i-1] = nieko


         then dėmuo := dėmuo*10 + i


         else begin


                reikšmė := reikšmė + ženklas*dėmuo;


                dėmuo := i;


                if oper[i-1] = plius


                   then ženklas := 1


                   else ženklas := -1;


               end;


    skaičiuoti := reikšmė + ženklas*dėmuo


  end; { skaičiuoti }





  procedure spausdinti (oper: lentelė);


    { išspausdina rastąjį reiškinį }


    var i: integer;


  begin


    write ('1');


    for i := 2 to MAX do


      case oper[i-1] of


        nieko: write (i);


        plius: write ('+', i);


        minus: write ('-', i);


      end;


    writeln;


  end; { spausdinti }





  procedure formuoti_skaičių (k: integer);


    { rekursinė procedūra formuojanti visus galimus reiškinius }


    var op: operacijos;


  begin


   if k = MAX { jei pasiekėme paskutinį skaitmenį }


     then begin


            if skaičiuoti (oper) = n { jei gavome tai ko reikia }


               then begin


                      rasta := true; { spausdiname rezultatą }


                      spausdinti (oper);


                    end;


          end


     else for op := plius to nieko do { bandome visus variantus }


            begin


              oper[k] := op;


              formuoti_skaiciu (k+1)


            end;


  end; { formuoti_skaičių }





begin


  readln (n);


  rasta := false; { nė vieno sprendinio nerasta }


  formuoti_skaičių (1);


  if not rasta


     then writeln ('NĖRA SPRENDINIO');


end.








178. Užrašas spirale. Pateikiame vieną galimų sprendimų. 


Sąlygoje nurodyta vietoje tarpų spausdinti pliusus. Tad pirmas darbas perskaičius sakinį: jame esančius tarpus pakeisti pliusais. 


procedure tarpai_pliusai (var e: string);


  { tarpus eilutėje pakeičia simboliais „+“ }


  var vieta: integer;


begin


  while pos (' ', e) > 0 do


  { kol sakinyje likęs nors vienas tarpas }


    begin


      vieta := pos (' ', e);


      e := copy (e, 1, vieta-1) +


           '+' +


           copy (e, vieta+1, length(e)-vieta)


    end;


end; { tarpai_pliusai }





Spirale užrašytą sakinį įrašysime į stačiakampę lentelę. Pastebėkite, kad kiekviena spiralės kraštinė pailgėja vienetu.

Sakinio ilgis neviršija 256 simbolių. Taigi, bet kokį sakinį galime sutalpinti į 25( 25 dydžio kvadratą. Pasirenkame šio kvadrato vidurinį langelį ir keisdami kryptis (žemyn, dešinėn, aukštyn, kairėn, žemyn, ir t. t.) įrašinėjame į kvadratą sakinį po vieną raidę. Kiekvieną kartą nauja spiralės kraštinė padidinama vienu simboliu (žr. pav.). 


const MAX = 25; { maksimali kvadrato, į kurį telpa spiralė, kraštinė }


type lentelė = array [1..MAX, 1..MAX] of char;


     kryptis = (žemyn, dešinėn, aukštyn, kairėn);





procedure rašyti (sk: integer; kr: kryptis; e: string;


                  var eil, st: integer;


                  var lent: lentelė);


{ įrašo į kvadratą sk sakinio simbolių nurodyta kryptimi, pradedant langeliu,


   kurio koordinatės yra (eil, st) }


  var i: integer;


begin


  if sk > length (e)


     then sk := length (e);


  for i := 1 to sk do


    begin


      lent[eil, st] := e[i];


      case kr of


        žemyn:   eil := eil + 1;


        aukštyn: eil := eil - 1;


        kairėn:  st  := st - 1;


        dešinėn: st  := st + 1;


      end;


    end;


end; { rašyti }





function succ_kr (kr: kryptis): kryptis;


begin


  if kr = kairėn


     then succ_kr := žemyn


     else succ_kr := succ (kr);


end; { succ_kr }





function tuščia (lent: lentelė; nr: integer;


                 eilutė: boolean): boolean;


{ patikrina, ar nurodyta lentelės eilutė (stulpelis), kurios numeris nr 


   yra sudaryta vien iš tarpų }


  var tarpai: boolean;


      i: integer;


begin


  tarpai := true;


  for i := 1 to MAX do


    if eilutė


       then tarpai := tarpai and (lent[nr, i] = ' ')


       else tarpai := tarpai and (lent[i, nr] = ' ');


  tuščia := tarpai


end; { tuščia }





procedure stumti (var lent: lentelė;


                  var eil_sk, st_sk: integer 


                  { eilučių bei stulpelių skaičius });


{ spiralę pastumia į lentelės kairįjį viršutinį kampą ir suskaičiuoja kiek


   eilučių bei stulpelių turi minimalus stačiakampis į kurį telpa spiralė }


  var i, j, te, ts: integer;


begin


  { randame tuščių eilučių skaičių lentelės viršuje }


  te := 0;


  while tuščia (lent, te+1, true) do


    te := te + 1;


  { randame eilučių, reikalingų spiralei užrašyti, skaičių }


  eil_sk := 0;


  while not tuščia (lent, eil_sk+te+1, true) do


    eil_sk := eil_sk + 1;


  { randame tuščių stulpelių skaičių lentelės kairėje }


  ts := 0;


  while tuščia (lent, ts+1, false) do


    ts := ts + 1;


  { randame stulpelių, reikalingų spiralei užrašyti, skaičių }


  st_sk := 0;


  while not tuščia (lent, st_sk+ts+1, false) do


    st_sk := st_sk + 1;


  { perkeliame spiralę į lentelės kairįjį viršutinį kampą }


  for i := 1 to eil_sk do


    for j := 1 to st_sk do


      lent[i, j] := lent[i+te, j+ts]


end; { stumti }





procedure užrašyti_spirale (e: string; { duotasis sakinys }


                            var lent: lentelė;


                            var eil_sk, st_sk: integer


                            { eilučių bei stulpelių, reikalingų


                             spiralei užrašyti, skaičius });


  var kiek, simb, i, j, eil, st: integer;


      kr: kryptis;


begin


  { sakinyje tarpus keičiame simboliais „+“ }


  tarpai_pliusai (e);


  { lentelę užpildome tarpais }


  for i := 1 to MAX do


    for j := 1 to MAX do


      lent[i, j] := ' ';


  { parenkame pradinį langelį }


  eil := MAX div 2; st := MAX div 2;


  simb := length (e); { simbolių kiekis sakinyje }


  kiek := 1;          { simbolių skaičius spiralės kraštinėje }


  kr := žemyn;        { pradinė kryptis }


  while simb > 0 do


  { kol neužrašėme viso sakinio }


    begin


      rašyti (kiek, kr, e, eil, st, lent);


      delete (e, 1, kiek); { pašalinama įrašytoji sakinio dalis }


      simb := simb - kiek; { sutrumpėja sakinio ilgis }


      kr := succ_kr (kr);


      kiek := kiek + 1


    end;


  { pastumsime spiralę į pradžią }


  stumti (lent, eil_sk, st_sk);


end; { užrašyti spirale }

